Advanced Vibration
A two d.o.f system with viscous damping

```
/BATCH,LIST
/COM,
/PREP7
/TITLE
ET,1,COMBIN40          ! DISPLACEMENT ALONG X AXIS, MASS AT NODE I
R,1,200.,0.4,1.0
R,2,100.,0.1,2.0       ! Ki, Ci AND Mi
R,3,200.,0.6,0.0
MP,EX,1,1              ! NOT USED, DUMMY PROPERTY
N,1                 ! DEFINE MODEL
N,2,1
N,3,2
N,4,3
E,2,1
REAL,2
E,3,2
REAL,3
E,4,3
D,1,UX,0.0             ! CONSTRAINT THE BASE
D,4,UX,0.0
OUTPR,ALL,ALL
FINISH
```



$k_1=200^{N/m}$, $k_2=100^{N/m}$, $k_3=200^{N/m}$

$c_1=0.4^{N.s/m}$, $c_2=0.1^{N.s/m}$, $c_3=0.6^{N.s/m}$

$m_1=1^{Kg}$, $m_2=2^{Kg}$

```
/SOLU
ANTYPE,MODAL           ! PERFORM A MODAL ANALYSIS
MODOPT,SUBSP,2         ! SUBSPACE IERTATION METHOD
                    ! EXTRACT 2 MODES FROM ENTIRE FREQUENCY RANGE
MXPAND,2,,,YES         ! EXPAND 2 MODES
SOLVE
*GET,FREQ1,MODE,1,FREQ
*GET,FREQ2,MODE,2,FREQ
FINISH

*DIM,LABEL,CHAR,2,2
*DIM,VALUE,,4,3
LABEL(1,1) = 'wn1,     ','wn2,     '
LABEL(1,2) = '  rad/s','  rad/s'
*VFILL,VALUE(1,1),DATA,11.0418,18.1129
*VFILL,VALUE(1,2),DATA,(FREQ1*6.2832),(FREQ2*6.2832)
*VFILL,VALUE(1,3),DATA,ABS(FREQ1*6.2832/11.0418),ABS(FREQ2*6.2832/18.1129)
/COM
/COM,------------------- Example RESULTS COMPARISON --------------
/COM,
/COM,              |  TARGET  |  ANSYS  |  RATIO
/COM,
*VWRITE,LABEL(1,1),LABEL(1,2),VALUE(1,1),VALUE(1,2),VALUE(1,3)
(1X,A8,A8,'   ',F10.4,'  ',F10.4,'   ',1F6.4)
/COM,--------------------------------------------------------------
FINISH
```

RESULTS:

```
------------------- Example RESULTS COMPARISON --------------

              |  TARGET  |  ANSYS  |  RATIO

wn1,     rad/s      11.0418     11.0419   1.0000
wn2,     rad/s      18.1129     18.1130   1.0000


--------------------------------------------------------------------------------
```